
News

Signal version 4.08 released
The latest revision of Signal is now available and includes the following new features:

• The clamp control bar can either display total electrode resistance or access and 
membrane resistance

• The Magstim control system now detects the version of firmware used in Rapid 
systems and will work with all versions.

For a full list of new features, see the revision history in the on-line help. 

Future meetings and events
Society for Neuroscience 2010
San Diego Convention Center,
San Diego, CA, USA
November 13th – 17th  2010

If you are organising or attending a conference at which you feel CED would be interested in 
exhibiting please let us know about it by notifying sales@ced.co.uk

US Training Days at SfN
Our US training days are now confirmed for the 11th and 12th of November at the Holiday 
Inn San Diego - On The Bay Hotel, prior to the Society for Neuroscience annual meeting. 
The training days are offered at $250 per day, including buffet lunch, materials and coffee 
and tea, and are suitable for both existing and prospective users of Spike2 and Signal. 
Places are filling up quickly so if you or any of your colleagues would like to attend, please 
follow the links to view the training day program and complete the registration form. Please 
give a short description of the type of research that you do and analysis that you are 
interested in. This will allow us to tailor sessions for specific requirements.

Latest versions of Spike2 and Signal
Updates for Spike2 and Signal are available from the CED downloads page, or by clicking on 
the links in the table below. Demonstration versions of the latest software are also available. 

 Spike2 downloads Released Signal downloads Released
Spike2 version 7.04 07/10 Signal version 4.0  8   08/10
Spike2 version 6.1  5  08/10 Signal version 3.13 05/10

Spike2 demo Signal demo

Q. I am using PCA (principal component analysis) clustering to sort my spikes and 
occasionally see what appears to be two separate clusters for what I believe to be a 
single spike. There is a change in the shape of the spike following addition of a drug, is 
there any way for me to tell if this cluster does indeed belong to the same spike?

A.  It may help if you were to use the time range option in the clustering dialog. Under the 
View menu there is the option to Show Time Range. If you show this and check the 
Enable box it will only display the dots relating to a specified percentage of the time 
range. There is then a slider which allows you to move through the file showing how 
the dots are positioned through time. If you see a cluster moving position, or one 
cluster disappearing as another appears, this could indicate a change of shape of the 
same spike. 

Another useful feature is the option to view event times as the z axis. If you view the 
clusters along the X axis you can align the view so that you can see changes in the 
principal components through time. 

From version 7.04 there are new toolbar buttons within the clustering window to give 
you fast access to the following useful features: Auto-scale, Link items 
closer than the minimum time, Display the axes, Use event times 
as the Z axis and Display as density plot. 

Cluster dialog with the z axis as time and viewed as density plot. Example shows cluster  
shapes changing through time

Q. I have two single frame average files containing responses before and after adding a 
drug to the preparation. Is there a way in Signal to subtract one average from another?

A. What you need to do is to get the averages into the same file in different frames. The 
following solution assumes you are working from saved average files rather than 
averages just created from an open source data file. Open both averages then select 
one and press Ctrl+C to copy the data into the clipboard. Click on the other average 
and, from the analysis menu select ‘Append frame’ to add an empty frame to the file. 
Press Ctrl+V and it will paste the clipboard data into the new frame, leaving you with 
a file containing both averages. 

From this point you can subtract one from another using the buffer. Ideally you would 
want to end up with a file containing both initial averages in the first two frames and the 
subtracted result in the third, so select the frame from which you wish to subtract the 
other and from the analysis menu use Append copy which will put a copy of this into 
a new frame at the end of the file. Go to the frame you wish to subtract and copy to the 
buffer (either from the analysis menu or by pressing the Ins key). You can then go to 
the third frame and subtract the buffer data (either from the analysis menu or simply 
by pressing the – key). The steps are illustrated below.  

Scripts: Spike2 

Q. I have an experimental preparation where I need to stimulate two sites and record from 
a third. I would like to be able to stimulate with a set number of repeats at varying inter-
stimulus intervals with fixed step sizes but randomising the order in which they are 
presented. A typical example would be 10 repeats each of intervals between -10ms 
and +10ms at 2ms steps giving 110 stimulus sets in total.

A. The attached script RandPairPulse.s2s and output sequence 
RandPairPulse.pls provide this functionality. First you must load the output 
sequence into your sampling configuration. When you then run the script this will 
control the sampling. The script will initially open a new data file ready to sample and 
has a toolbar with buttons from which you control it.  

At any time the file is open (either before pressing Sample start or when it is 
sampling) you can press Pulse settings to set up the stimulus parameters. 
Pressing the Stimulate button during sampling will start the output. You can test it by 
simply connecting the digital outputs to the associated event inputs and recording 
these in a couple of event+ channels.  

When the pulses have stopped, if you press Stimulate again it will generate the 
same randomised pattern, however if you wish to change the order just redo Pulse 
Settings and it will generate the same intervals in a different random order.

CED
Hint: You can achieve this in Signal by setting multiple states with different intervals and setting the state ordering to random

CED
Hint: Signal data can be copied and pasted easily between files. Ctrl+C will add any visible data to the clipboard and this can be pasted to other frames within the same file, to other files and even to XY views.


'$RandPairPulse.s2s|Spike2 script to generate randomly presented sets of paired pulses
'Requires RandPairPulse.pls output sequence to run

'User specifies an inital interval between pulses on digital outputs 0 and 1
'   together with an interval step and number of steps and repeats 
'Typical example would generate intervals between 10ms and -10ms in 2ms steps, with 10 repetitions of each
'These are then presented in a randomised order.

'Current version uses 2 second intervals between pairs of pulses. 



var data%;                                                      'Data file handle
var initInt% := 10;                                             'Initial interval between pulses on dig out 0 and dig out 1
var stepInt% := 2;                                              'Interval decrease per step
var nSteps% := 10;                                              'Number of decreasing interval steps
var nReps% := 10;                                               'Number or repetitions at each interval
var nPairs%;                                                    'Total number of pulse pairs


HideAll();                                                      'Hide all unwanted bars
ToolbarVisible(1);                                              'Make toolbar visible always
New%();                                                         'Set up new sampling window
DoToolbar();                                                    'Do the toolbar
RestoreAll();                                                   'Restore all bars
Halt;


proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);                                    'Set up toolbar buttons
ToolbarSet(2,"Sample start", Start%);
ToolbarSet(3,"Sample stop", Stop%);
ToolbarSet(4,"New file", New%);
ToolbarSet(6,"Pulse Settings", SetPulses%);
ToolbarSet(8,"Stimulate",DoPulses%);
ToolbarEnable(3,0);                                             'Disable "Sample stop" button 
ToolbarEnable(8,0);
Toolbar("Press SAMPLE START to commence sampling", 1023);       'Wait here until quit
end;



func New%()                                                     'New sampling window
if data%>0 then                                                 'If there is already a data view open...
  View(data%);                                                  '...close it
  FileClose();                                                  'No point setting data% to 0...
endif;

data%:=FileNew(0,3);                                            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
Window(0,0,100,100);                                            'Make data window in top bit of screen
XRange(0,10);
FrontView(data%);                                               'Bring the data view to the front
ToolbarEnable(3,0);                                             'Disable "Sample stop" button
ToolbarEnable(2,1);                                             'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;



Func SetPulses%()
var ok%;
var i%,j%;
DlgCreate("Randomised paired pulse stimulation");
DlgInteger(1, "Initial interval (Dig 0 to Dig 1) (ms)",1,100);
DlgInteger(2, "Interval step (ms)",1,10);
DlgInteger(3, "Number of steps",1,100);
DlgInteger(4, "Repetitions at each interval",1,100);
ok% := DlgShow(initInt%,stepInt%,nSteps%,nReps%);
if ok% > 0 then
    nPairs% := (nSteps%+1) * nReps%;
    var intervals%[nPairs%];                                    'Array to contain list of intervals
    var randArr[nPairs%];                                       'Array to fill with random numbers for randomisation of order
    var seqTable%[nPairs% * 2];                                 'Array of integer values to pass to sequencer table
    ArrConst(intervals%[:nReps%],initInt%);                     'Set first block of repeats to initial interval
    j% := nReps%;
    for i% := 1 to nSteps% do
        ArrConst(intervals%[j%:nReps%],initInt% - (i%*stepInt%)); 'Set each subsequent block with decremented interval
        j% += nReps%;
    next;
    Rand(randArr[]);                                            'Fill array with random values
    ArrSort(randArr[],0,intervals%[]);                          'Sort random array with interval array linked. Will randomise interval array 
    for i% := 0 to nPairs%-1 do                                 'For each pair
        docase
        case intervals%[i%] < 0 then                            'If dig 1 precedes dig 0
            seqTable%[i%*2] := 0;                               '   Set table jump value to zero
            seqTable%[(i%*2)+1] := ((0-intervals%[i%])*10)-4;   '   Set table interval in sequencer steps
        case intervals%[i%] = 0 then                            'If simultaneous outputs
            seqTable%[i%*2] := 1;                               '   Set table jump value to 1
            seqTable%[(i%*2)+1] := 0;                           '   Set table interval value to zero
        case intervals%[i%] > 0 then                            'If dig 0 precedes dig 1
            seqTable%[i%*2] := 2;                               '   Set table jump value to 2
            seqTable%[(i%*2)+1] := (intervals%[i%]*10)-4;       '   Set table interval in sequencer steps
        endcase;
    next;
    SampleSeqTable(seqTable%[]);                                'Pass array to sequencr table
    SampleSeqVar(2,(nPairs%-1)*2);                              'Pass value set number of pairs to sequence
    ToolbarEnable(8,1);                                         'Enable Stimulate button
endif;

return 1;
end;


Func  DoPulses%()
SampleKey("g");
return 1;
end;


func Start%()                                                   'If "Start" is pressed
SampleStart();                                                  'Start sampling
ToolbarEnable(4,0);                                             'Disable "New file" button
ToolbarEnable(3,1);                                             'Enable "Sample stop" button
ToolbarEnable(2,0);                                             'Disable "Sample start" button
ToolbarEnable(1,0);                                             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                                                       'Stay with toolbar
end;                                
                                
                                
func Stop%()                                                    'If "Stop" is pressed
SampleStop();                                                   'Stop sampling
if (data%>0) and                                                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);                                             'Enable "New file" button
ToolbarEnable(3,0);                                             'Disable "Sample stop" button
ToolbarEnable(1,1);                                             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                                                       'Stay in toolbar
end;                                
                                
func Quit%()                                                    'If "Quit" is pressed
SampleStop();                                                   'Stop sampling
return 0;                                                       'leave toolbar
end;


'You can find this code in the App() on-line help.
var gFloat%[20];                                                'global for floating window states
proc HideAll()                              
var i%;                             
gFloat%[0] := App(-2);                                          'number of windows
for i% := 1 to gFloat%[0] do                                    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do                                    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

CED
RandPairPulse.s2s




CED
RandPairPulse.pls



Scripts: Signal 

Q: I record long sweeps of 100 seconds in Signal and would like to display the last 10 
seconds during recording with the data scrolling as it does in Spike2, is this possible? 

A: The attached script  ScrollLongFrames.sgs  uses an idle  routine to  update the 
view as often as possible to create this effect. If the displayed x axis range is shorter 
than the frame length it will update the displayed x axis range to show the current time  
range up to the last available sampled data point. You can toggle the scrolling function 
on and off using a toolbar button created by the script. 

Did you know…?

The  maximum  frame  length  in  Signal  is  related  to  1401  memory.  As  a  general  rule, 
approximately 90% of the 1401 memory can be used within a single sweep in most sampling 
modes. However,  for fast  sweep modes this memory is split  into 4 parts so the possible 
sweep length is reduced accordingly. 

When using a Power1401 with a large amount of memory, consideration must then be given 
to the available RAM in the host computer. The data file records 2 bytes per point, however 
when Signal displays this it actually requires 10 bytes per point, therefore trying to display 
very long sweeps holding a large amount of data can cause problems.   

Recent questions

Q. I have recorded a signal which I  would like to play through the sound card of  my 
computer. I have tried this but the volume produced is very low even with the speaker 
volume turned up high. Is there a way to increase this?

A. The amplitude of the output waveform is proportional to the ADC range taken up by 
the signal. You can get an idea of the overall size of your signal by showing the Y 
range dialog for the channel and pressing  Show All.  If  this shows the signal is 
actually quite small,  a simple way to rescale the signal so that it  fills the available 
range is to apply an all pass FIR filter and ensure the Rescale box is checked. If you 
then output this channel you should find it is much louder. 

Digital filter application dialog

User group     

You can now find our eNewsletters archived on the CED website.

Try the CED Forums bulletin board for software and hardware support

If you have any comments about the newsletter format and content, or wish to un-subscribe  
from the mailing list for this newsletter, please notify sales@ced.co.uk.

'$ScrollLongFrames.sgs|Signal script to enable scrolling during long recording frames
' This script samples into a time view, allowing the user to scroll the window if 
'   the displayed time range is shorter than the frame length
' It uses the existing sampling configuration and takes control of the sampling via a 
'   script generated toolbar
'When the script is running there is a 'Scroll on/Scroll off button. When scrolling is on
'   it calculates the current displayed time range and using an idle routine updates the view
'   to show this range back from the last sampled data point.

var sv%;								' The file view handle
var ch%, pt%;							' Number of channels and pointer per channel
var fLen;                               ' Frame length
var vLen;                               ' Duration to display
var scrollFlag% := 0;
var ports%[80];							' The sampling ports list
var amax[80];							' Pre-calculated limit values
var amin[80];							' so that we run faster

View(App(3)).WindowVisible(0);			' Hide the script
SetParameters();						' Clean up sampling parameters
SetToolbar();							' Set up the toolbar
ToolBar("Sampling with artefact rejection", 1023);	' Do it
View(App(3)).WindowVisible(1);			' Show the script again
halt;



'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Set up the sampling parameters so that they don't give problems. Mostly, use
' whatever parameters are current.
'
proc SetParameters()
SampleWrite(0);							' Sweeps not written to disk
SamplePause(1);							' Pause after each sweep
return;
end;

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
proc SetToolbar()
ToolbarClear();
ToolbarSet(0, "", Idle%);
ToolbarSet(4, "Start", Start%);
ToolbarSet(3, "Stop", Stop%);
ToolbarSet(2, "Abort", Abort%);
ToolbarSet(1, "Quit", Quit%);
ToolbarSet(6, "Scroll on", DoScroll%);
ToolbarEnable(3, 0);						' At the start, no Stop or Abort
ToolbarEnable(2, 0);
return;
end;

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Idle does most of the work. If there is a new sweep available we check for an
' artefact and, if OK, add the sweep into the file.
'
func Idle%()
var s%, i%, v;
s% := SampleStatus();
if (s% > 0) then
    vLen := XHigh()-XLow();
    if scrollFlag% > 0 then
    if s% = 2 then
        if MaxTime() < (vLen + MinTime()) then
            Draw(MinTime(),vLen);
            'ToolbarText(Print$("%f  still",MaxTime()));
        else
            Draw(MaxTime()-vLen,vLen);
            'ToolbarText(Print$("%f",MaxTime()));
        endif;
        endif;
    endif;
    
   if (s% = 3) then						' If we have got a sweep available
      SampleSweep();					' Enable a new sweep
   endif;
endif;
return 1;
end;


'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Start - create the time and result views, initialise the views and position
'         them on the screen and initiate the sampling process. We also
'         pre-calculate the ADC limit values to save time.
'
func Start%()
var er$, sh%;
var p%, abit, azero, c%;
sv% := FileNew(0, 3);					' Create a file view for sampling
if (sv% < 0) then						' and check for errors
   er$ := Error$(sv%);
   Message("File new failed :\n\n%s", er$);
   return 0;
   endif;
sh% := SampleHandle(0);
if (sh% <> sv%) then
    Message("Handle %d should be %d\n", sh%, sv%);
	return 0;
	endif;

' Now get some information about the sampled data and calculate limits
ch% := SamplePorts(ports%[]);			' Get # waveform channels and port list
pt% := SamplePoints();					' and the points per channel
fLen := pt%/SampleRate();
vLen := XHigh()-XLow();
for c% := 1 to ch% do					' Now calculate the limit values
  p% := ports%[c% - 1];					' The port for this channel
  amax[c%] := SamplePortFull(p%);		' Max ADC value
  azero := SamplePortZero(p%);			' Midrange ADC value
  amin[c%] := azero - (amax[c%] - azero);	' Minimum value
next

ToolbarEnable(4, 0);					' Turn off Start button
ToolbarEnable(3, 1);					' Turn on Stop and Abort
ToolbarEnable(2, 1);
View(sv%);								' File view is current view
Frame(0);								' Fix file view on frame 0
SampleStart();							' and start sampling
return 1;
end;


'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Stop - stop the sampling and leave the file view filling the screen.
'
func Stop%()
SampleStop();							' Stop all sampling
ToolbarEnable(4, 1);					' Turn on Start button
ToolbarEnable(3, 0);					' Turn off Stop and Abort
ToolbarEnable(2, 0);
return 1;
end;

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Abort - abort sampling, which will kill off the file view.
'
func Abort%()
SampleAbort();
ToolbarEnable(4, 1);					' Turn on Start button
ToolbarEnable(3, 0);					' Turn off Stop and Abort
ToolbarEnable(2, 0);
return 1;
end;

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Quit - return zero to terminate ToolBar function, and hence the
'	     execution of the script.
'
func Quit%()
return 0;
end;

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
'Scroll on - enables scrolling of data 
Func DoScroll%()
scrollFlag% := 1;
ToolbarSet(6, "Scroll off", NoScroll%);
return 1;
end;


'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
'Scroll off - enables scrolling of data 
Func NoScroll%()
scrollFlag% := 0;
ToolbarSet(6, "Scroll on", DoScroll%);
return 1;
end;

CED
ScrollLongFrames.sgs




